LHeC electron ring: parameter choices

John Jowett

For LHeC workshop, Divonne, 2 September 2008.

Much already done in Dainton et al 2006, try to review some choices and fix some more parameters.

Setup

Fixed basic parameters

In[7]:=
```
CircLHC = 26 658.8832 Meter
```

Out[7]=
```
26 658.9 Meter
```

In[8]:=
```
LHCbasic = {frev -> Convert[c/CircLHC // N, Hertz]}
```

Out[8]=
```
{frev → 11 245.5 Hertz}
```

LEP optics for comparisons

Some items taken from LEP Design Report

In[9]:=
```
NcellLEP = 8 × 31
```

Out[9]=
```
248
```

In[10]:=
```
LEPoptics = tfsRead["G:\\Users\\j\\jowett\\Documents\\Private\\Other
      Accelerators\\LEPpostfacto\\LEP.tfs"] /. "K0L" → "ANGLE";
```

In[11]:=
```
First[LEPoptics] // TableForm
```

From LEP Design Report p 6

In[12]:=
```
nel = Length[Last[LEPoptics]]
```

Out[12]=
```
9264
```

Distances between quadrupoles

In[13]:=
```
Take[mfsColumn[mfsMember[LEPoptics, "KEYWORD", {"QUADRUPOLE"}], "NAME"], 50]
```

In[14]:=
```
LEPcell = mfsOpticsRange[LEPoptics, {"QF33.R1", "QF35.R1"}];
```

In[15]:=
```
s0LEPcell = First[mfsColumn[LEPcell, "S"]]
```

Out[15]=
```
1034.98
```

In[16]:=
```
Options[BeamLineElementGraphic]
```
Out[16]=
```
{BeamLineGraphicWidth → 1, BeamLineGraphicDot → 0.3, BeamLineGraphicLabelLength → 0.5}
```
In[17]:=
```
SetOptions[BeamLineElementGraphic,
  BeamLineGraphicWidth → 15, BeamLineGraphicLabelLength → 1];
mfsOpticsClassicBeamLinePlot[LEPcell, AspectRatio → .5, DxFactor → 100,
 PlotRange → {0, 200}, BeamLineGraphicVerticalShift → 130]
```
Out[18]=



The phase advances in this cell:

In[19]:=
```
With[{mux = mfsColumn[LEPcell, "MUX"], muy = mfsColumn[LEPcell, "MUY"]},
  {Last[mux] - First[mux], Last[muy] - First[muy]}] 360
```
Out[19]=
```
{91.0332, 59.436}
```

So this is the classic 90°-60° optics.

In[20]:=
```
squads = mfsColumn[mfsMember[LEPoptics, "KEYWORD", {"QUADRUPOLE"}], "S"];
```

Spaces between quads in LEP. Most common case should be length of half-cell.

In[21]:=
```
ListPlot[squads - RotateRight[squads]]
```

Out[21]=



In[22]:=
```
Select[Tally[squads - RotateRight[squads]], #1⟦2⟧ > 100 &];
LFODOLEP = 2 First[First[%]]
```

Out[23]= 79.

Get bending radius from a typical bend magnet

In[24]:=
```
rhoLEP = "L"/"ANGLE" /. mfsToRules[LEPoptics, "B2R.QF19.R1"]
```

Out[24]= 3065.21

In[25]:=
```
LquadLEP = "L" /. mfsToRules[LEPoptics, "QD34.R1"]
```

Out[25]= 1.6

This is the magnetic length of the 6-core dipole.

In[26]:=
```
LbendLEP = 35.01
```

Out[26]= 35.01

The physical length is a bit different

In[27]:=
```
LbendLEP1 = 6 × 5.75
```

Out[27]= 34.5

In[28]:=
```
LbendLEP/LFODOLEP
```

Out[28]= 0.443165

In[29]:=
```
LquadLEP/LFODOLEP
```

Out[29]= 0.0202532

In[30]:=
```
LsextFLEP = 0.4; LsextDLEP = 0.76;
```

How much space is left for the other items

In[31]:=
```
LFODOLEP - 2 LbendLEP - 2 LquadLEP - LsextDLEP - LsextFLEP
```

Out[31]= 4.62

In[32]:=
```
2 LbendLEP/LFODOLEP
```
Out[32]= 0.886329

In[33]:=
```
NcellLEP LFODOLEP
```
Out[33]= 19 592.

The LEP vacuum chamber is elliptical, with half-axes:

In[34]:=
```
LEPvacuumChamber = {131., 70.}/2 Milli Meter
```
Out[34]= {65.5 Meter Milli, 35. Meter Milli}

Rough estimate of σ in the arcs

In[35]:=
```
Convert[√(40 Nano Meter 150 Meter) // N, Milli Meter]
```
Out[35]= 2.44949 Meter Milli

In[36]:=
```
ToFundamentalSI[N[131 Milli Meter/√(40 Nano Meter 150 Meter)]]
```
Out[36]= 53.4805

Allowing for sawtooth, pretzels, closed-orbit

In[37]:=
```
Convert[10 √(40 Nano Meter 150 Meter) + 25 Milli Meter // N , Milli Meter]
```
Out[37]= 49.4949 Meter Milli

That more or less accounts for it.

For LHeC, this would change roughly to

In[38]:=
```
Convert[10 √(8 Nano Meter 150 Meter) + 10 Milli Meter // N , Milli Meter]
```
Out[38]= 20.9545 Meter Milli

LHC arc optics for comparison

In[39]:=
```
NcellLHC = 8 × 23
```
Out[39]= 184

In[41]:=
```
LHCcell = tfsRead["LHCB1ArcCell.tfs"] /. "K0L" → "ANGLE";
```

In[43]:=
```
LFODOLHC = With[{svals = mfsColumn[LHCcell, "S"]},
  -Min[svals] + Max[svals]]
```
Out[43]= 106.903

In[45]:=
```
LFODOLEP/LFODOLHC
```
Out[45]= 0.738988

In[46]:=
```
SetOptions[BeamLineElementGraphic,
  BeamLineGraphicWidth → 15, BeamLineGraphicLabelLength → 1];
mfsOpticsClassicBeamLinePlot[LHCcell, AspectRatio → .5, DxFactor → 100,
 PlotRange → {0, 200}, BeamLineGraphicVerticalShift → 130]
```

Out[47]=



FODO arc cells

No reason to choose anything else.

See for example Chao-Tigner handbook, Sec 2.2.3 by E Keil

■ Tentative choice of cell length for LHeCe

Show the two cells, LHC and LEP, beside each other. Would it not make the engineering and integration easier if there was a simpler relation between the LHC and LHeCe rings ?

```
In[48]:= Show[Graphics[{Translate[mfsOpticsBeamLineGraphics[LEPcell], {-s0LEPcell, 50}],
            mfsOpticsBeamLineGraphics[LHCcell]}]]
```

Out[48]=



For example, we could choose the LHeCe FODO cell to be half the length of the LHC cell. This would give a number of cells

```
In[49]:= 2 NcellLHC
Out[49]= 368

In[50]:= NcellLEP
Out[50]= 248
```

Such a choice would also impose some constraints on insertions, bypasses and their matching. We leave those to later.

```
In[51]:= Notation[L_FODO ⟺ LFODO];
         IntroduceSymbol[LFODO, "is the half-length of the FODO cell.", Meter];

In[53]:= Notation[φ_FODO ⟺ phiFODO];
         IntroduceSymbol[phiFODO, "is the bending angle of a whole FODO cell.", 1];

In[55]:= Notation[N_cell ⟺ Ncell];
         IntroduceSymbol[Ncell, "is the number of cells in the ring.", 1];
```

In[57]:= $\texttt{FODOphi} = \varphi_{\texttt{FODO}} == \dfrac{2\,\pi\,\mathrm{L}_{\texttt{FODO}}}{\texttt{NcellLHC LFODOLHC Meter}}$

Out[57]= $\varphi_{\texttt{FODO}} == \dfrac{0.000319427\,\mathrm{L}_{\texttt{FODO}}}{\texttt{Meter}}$

In[58]:= `FODONcell = ` $N_{cell} == \frac{\text{NcellLHC LFODOLHC}}{L_{FODO}}$

Out[58]= $N_{cell} == \frac{19\,670.1}{L_{FODO}}$

In[59]:= `halfLHCcell = ` $\left\{\text{LFODO} \to \frac{\text{LFODOLHC Meter}}{2}, \text{Ncell} \to 2\,\text{NcellLHC}, \varphi_{FODO} \to \frac{2\pi}{2\,\text{NcellLHC}}\right\}$

Out[59]= $\left\{L_{FODO} \to 53.4515\,\text{Meter}, N_{cell} \to 368, \varphi_{FODO} \to \frac{\pi}{184}\right\}$

In[60]:= `fullLHCcell = ` $\left\{\text{LFODO} \to \text{LFODOLHC Meter}, \text{Ncell} \to \text{NcellLHC}, \varphi_{FODO} \to \frac{2\pi}{\text{NcellLHC}}\right\}$

Out[60]= $\left\{L_{FODO} \to 106.903\,\text{Meter}, N_{cell} \to 184, \varphi_{FODO} \to \frac{\pi}{92}\right\}$

■ Lengths in the cells

In[63]:=
```
Notation[ρFODO ⟺ rhoFODO];
IntroduceSymbol[rhoFODO, "is the bending radius of the thin lens FODO cell.", Meter];
```

In[65]:=
```
Notation[LS ⟺ Lsext];
IntroduceSymbol[Lsext, "is the average arc sextupole length.", Meter];
```

In[67]:=
```
Notation[Lother ⟺ Lother];
IntroduceSymbol[Lother,
  "is the length allowed for other items in a FODO cell.", Meter];
```

We can probably scale down the sextupole strengths by the ratio of maximum energy

In[223]:= `{LsextFLEP, LsextDLEP}`

Out[223]= `{0.4, 0.76}`

So let's take 1
m in total for the two sextpoles

In[70]:= `Lsext = 1 / 2 Meter; Lother = 4 Meter;`

Subtracting the two quadrupole lengths, and allow some space for sextupoles, BPMs, etc. will give the largest possible ρ

In[71]:= `FODOrho = ` $\rho_{FODO} == \frac{L_{FODO} - 2\,L_Q - 2\,L_S - L_{other}}{\varphi_{FODO}}$

Out[71]= $\rho_{FODO} == \frac{L_{FODO} - 5\,\text{Meter} - 2\,L_Q}{\varphi_{FODO}}$

Sometimes we can use the cruder approximation (which is about the value for LEP) in order to simplify certain results and remove weak dependences on L_Q.

In[72]:= `FODOrho0 = ` $\rho_{FODO} == \frac{9}{10}\,\frac{\text{LFODO}}{\text{phiFODO}}$

Out[72]= $\rho_{FODO} == \frac{9\,L_{FODO}}{10\,\varphi_{FODO}}$

Thin lens optics of FODO cells

In[73]:=
```
Notation[K_FODO ⟺ KFODO];
IntroduceSymbol[KFODO,
  "is the focusing strength of the FODO cell quadrupole.", Meter^-2];
```

In[75]:=
```
Notation[L_Q ⟺ Lquad];
IntroduceSymbol[Lquad, "is the length of the FODO cell quadrupole.", Meter];
```

In[77]:=
```
FODOfK = KFODO == 1/(fFODO Lquad)
```

Out[77]= $K_{FODO} == \frac{1}{fFODO\ L_Q}$

In[78]:=
```
Brho[70 GeV / c]
```

Out[78]= $\frac{70\ ElectronVolt\ Giga}{c\ e}$

In[79]:=
```
Notation[f_FODO ⟺ fFODO];
IntroduceSymbol[fFODO,
  "is the focal length of the focusing quadrupoles in the FODO cell.", 1/Meter];
```

In[81]:=
```
Notation[μ_FODO ⟺ muFODO];
IntroduceSymbol[muFODO, "is the phase advance of the FODO cell.", 1];
```

In[83]:=
```
Notation[β^+ ⟺ betap];
IntroduceSymbol[betap, "is the maximum β-function in the FODO cell.", Meter];
```

In[85]:=
```
Notation[β^- ⟺ betam];
IntroduceSymbol[betam, "is the minimum β-function in the FODO cell.", Meter];
```

In[87]:=
```
Notation[D_x^+ ⟺ Dxp];
IntroduceSymbol[Dxp, "is the maximum dispersion in the FODO cell.", Meter];
```

In[89]:=
```
Notation[D_x^- ⟺ Dxm];
IntroduceSymbol[Dxm, "is the minimum dispersion in the FODO cell.", Meter];
```

In[91]:=
```
FODOfmu = fFODO == LFODO / (4 Sin[muFODO/2])
```

Out[91]= $f_{FODO} == \frac{1}{4} L_{FODO} \operatorname{Csc}\left[\frac{\mu_{FODO}}{2}\right]$

In[92]:=
```
FODObetap = betap == LFODO (1 + Sin[muFODO / 2]) / Sin[muFODO]
```

Out[92]= $\beta^+ == L_{FODO} \operatorname{Csc}[\mu_{FODO}] \left(1 + \operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right)$

In[93]:=
```
FODObetam = betam == LFODO (1 - Sin[muFODO / 2]) / Sin[muFODO]
```

Out[93]= $\beta^- == L_{FODO} \operatorname{Csc}[\mu_{FODO}] \left(1 - \operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right)$

In[94]:= $$\texttt{FODODxp = Dxp == }\frac{\texttt{LFODO phiFODO}\left(1+\frac{1}{2}\,\texttt{Sin}\left[\frac{\texttt{muFODO}}{2}\right]\right)}{4\,\texttt{Sin}\left[\frac{\texttt{muFODO}}{2}\right]^2}$$

Out[94]= $$D_x^+ == \frac{1}{4} L_{FODO}\, \varphi_{FODO}\, \mathrm{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2 \left(1+\frac{1}{2}\mathrm{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right)$$

In[95]:= $$\texttt{FODODxm = Dxm == }\frac{\texttt{LFODO phiFODO}\left(1-\frac{1}{2}\,\texttt{Sin}\left[\frac{\texttt{muFODO}}{2}\right]\right)}{4\,\texttt{Sin}\left[\frac{\texttt{muFODO}}{2}\right]^2}$$

Out[95]= $$D_x^- == \frac{1}{4} L_{FODO}\, \varphi_{FODO}\, \mathrm{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2 \left(1-\frac{1}{2}\mathrm{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right)$$

In[96]:= `FODOOpticalFunctions := {FODOfK, FODOfmu, FODODxp, FODODxm, FODObetam, FODObetap}`

Make some plots

In[97]:= $$\frac{\texttt{betap}}{\texttt{LFODO}}\ \texttt{/. Solve[FODObetap, betap]}$$

Out[97]= $$\left\{\mathrm{Csc}[\mu_{FODO}]\left(1+\mathrm{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right)\right\}$$

In[98]:= `Solve[FODOOpticalFunctions, {Dxp, Dxm}]`

Out[98]= $$\left\{\left\{D_x^+ \to \frac{1}{8} L_{FODO}\, \varphi_{FODO}\, \mathrm{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2\left(2+\mathrm{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right),\ D_x^- \to -\frac{1}{8} L_{FODO}\, \varphi_{FODO}\, \mathrm{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2\left(-2+\mathrm{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right)\right\}\right\}$$

In[99]:= $$\texttt{First}\left[\left\{\frac{\texttt{betap}}{\texttt{LFODO}}, \frac{\texttt{betam}}{\texttt{LFODO}}\right\}\texttt{ /. Solve[FODOOpticalFunctions, \{betap, betam\}] /. muFODO} \to 2\,\pi\,\frac{\texttt{mudeg}}{360}\right]$$

Out[99]= $$\left\{\mathrm{Csc}\left[\frac{\mathrm{mudeg}\,\pi}{180}\right]\left(1+\mathrm{Sin}\left[\frac{\mathrm{mudeg}\,\pi}{360}\right]\right),\ \frac{L_{FODO}\,\mathrm{Csc}\left[\frac{\mathrm{mudeg}\,\pi}{180}\right] - L_{FODO}\,\mathrm{Csc}\left[\frac{\mathrm{mudeg}\,\pi}{180}\right]\mathrm{Sin}\left[\frac{\mathrm{mudeg}\,\pi}{360}\right]}{L_{FODO}}\right\}$$

In[100]:= $$\texttt{Plot}\Big[\texttt{Evaluate}\left[\texttt{First}\left[\left\{\frac{\beta^+}{L_{FODO}}, \frac{\beta^-}{L_{FODO}}\right\}\texttt{ /. Solve[FODOOpticalFunctions, }\{\beta^+, \beta^-\}\texttt{] /. }\mu_{FODO} \to \frac{2\,\pi\,\texttt{mudeg}}{360}\right]\right],$$
$$\texttt{\{mudeg, 0, 180\}, FrameLabel} \to \{\texttt{"}\mu/°\texttt{"}, \texttt{"}\beta^{\pm}_{\square}/L_{FODO}/m\texttt{"}\}\texttt{, PlotStyle} \to \texttt{\{Blue, Darker[Green]\}}\Big]$$

Out[100]=



In[101]:=
```
Plot[Evaluate@
  First[{Dxp, Dxm}/(phiFODO LFODO) /. Solve[FODOOpticalFunctions, {Dxp, Dxm}] /. muFODO → 2 π mudeg/360],
 {mudeg, 0, 180}, FrameLabel → {"μ/°", "D±/(φ L_FODO)/m"}, PlotStyle → {Blue, Darker[Green]}]
```

Out[101]=



If the cell starts at $s = 0$ with the QF

In[102]:=
```
Notation[β_FODO ⟺ betaFODO];
IntroduceSymbol[betaFODO, "is the beta function in the
   FODO cell starting at the QF (s=0) to the QD (s=L_FODO/2).", Meter];
```

In[104]:=
```
betaFODO[s_] = Module[{b1},
  b1 =
   (betap /. First@Solve[FODObetap, betap]) - 2 s (1 + Sin[muFODO / 2])/Cos[muFODO / 2] + 4 s^2/LFODO Tan[muFODO/2];
  Piecewise[{{b1, s/LFODO < 1/2}, {b1 /. s → (LFODO - s), 1/2 < s/LFODO < 1}}]]
```

Out[104]=

$$\begin{cases} L_{FODO} \operatorname{Csc}[\mu_{FODO}] \left(1 + \operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right) - 2 s \operatorname{Sec}\left[\frac{\mu_{FODO}}{2}\right] \left(1 + \operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right) + \frac{4 s^2 \operatorname{Tan}\left[\frac{\mu_{FODO}}{2}\right]}{L_{FODO}} & \frac{s}{L_{FODO}} < \frac{1}{2} \\ L_{FODO} \operatorname{Csc}[\mu_{FODO}] \left(1 + \operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right) - 2 (L_{FODO} - s) \operatorname{Sec}\left[\frac{\mu_{FODO}}{2}\right] \left(1 + \operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right) + \frac{4 (L_{FODO} - s)^2 \operatorname{Tan}\left[\frac{\mu_{FODO}}{2}\right]}{L_{FODO}} & \frac{1}{2} < \frac{s}{L_{FODO}} < : \end{cases}$$

In[105]:=
```
Notation[β^D_FODO ⟺ betaDFODO];
IntroduceSymbol[betaDFODO,
 "is the beta function in the vertical plane in the FODO cell
   starting at the QF (s=0) to the QD (s=L_FODO/2).", Meter];
```

In[107]:=
```
betaDFODO[s_] = Module[{b1},
  b1 = (betap /. First@Solve[FODObetap, betap]) -
     (2 s (1 + Sin[muFODO / 2]))/Cos[muFODO / 2] + (4 s^2)/LFODO Tan[muFODO/2] /. s → LFODO/2 - s;
  Piecewise[{{b1, s/LFODO < 1/2}, {b1 /. s → (LFODO - s), 1/2 < s/LFODO < 1}}]]
```

Out[107]=

$$\begin{cases} L_{FODO} \operatorname{Csc}[\mu_{FODO}] \left(1+\operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right) - 2\left(\frac{L_{FODO}}{2}-s\right)\operatorname{Sec}\left[\frac{\mu_{FODO}}{2}\right]\left(1+\operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right)+\frac{4\left(\frac{L_{FODO}}{2}-s\right)^2\operatorname{Tan}\left[\frac{\mu_{FODO}}{2}\right]}{L_{FODO}} & \frac{s}{L_{FODO}}<\frac{1}{2} \\ L_{FODO} \operatorname{Csc}[\mu_{FODO}] \left(1+\operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right) - 2\left(-\frac{L_{FODO}}{2}+s\right)\operatorname{Sec}\left[\frac{\mu_{FODO}}{2}\right]\left(1+\operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right)+\frac{4\left(-\frac{L_{FODO}}{2}+s\right)^2\operatorname{Tan}\left[\frac{\mu_{FODO}}{2}\right]}{L_{FODO}} & \frac{1}{2}<\frac{s}{L_{FODO}}<1 \end{cases}$$

In[108]:=
```
Notation[D_FODO ⟺ DFODO];
IntroduceSymbol[DFODO,
  "is the dispersion in the FODO cell starting at the QF (s=0) to the QD (s=L_FODO/2)..",
  Meter];
```

In[110]:=
```
DFODO[s_] = Module[{d1},
  d1 = (Dxp /. First@Solve[FODODxp, Dxp]) (1 - 2 s Sin[muFODO/2]/LFODO) + (s^2 phiFODO)/(2 LFODO);
  Piecewise[{{d1, s/LFODO < 1/2}, {d1 /. s → (LFODO - s), 1/2 < s/LFODO < 1}}]]
```

Out[110]=

$$\begin{cases} \frac{\varphi_{FODO}\, s^2}{2 L_{FODO}}+\frac{1}{8} L_{FODO}\,\varphi_{FODO} \operatorname{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2\left(2+\operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right)\left(1-\frac{2 s \operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]}{L_{FODO}}\right) & \frac{s}{L_{FODO}}<\frac{1}{2} \\ \frac{\varphi_{FODO}\,(L_{FODO}-s)^2}{2 L_{FODO}}+\frac{1}{8} L_{FODO}\,\varphi_{FODO} \operatorname{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2\left(2+\operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]\right)\left(1-\frac{2 (L_{FODO}-s) \operatorname{Sin}\left[\frac{\mu_{FODO}}{2}\right]}{L_{FODO}}\right) & \frac{1}{2}<\frac{s}{L_{FODO}}<1 \end{cases}$$

In[111]:=
```
Evaluate[{betaFODO[s]/LFODO, DFODO[s]/(LFODO phiFODO)} /. s → sl LFODO // Simplify]
```

In[112]:=
```
Evaluate[{betaFODO[s]/LFODO, DFODO[s]/(LFODO phiFODO)} /. {s → sl LFODO, muFODO → π / 3} // Simplify]
```

In[113]:=
```
Evaluate[{betaFODO[s]/LFODO, DFODO[s]/(LFODO phiFODO)} /. {s → sl LFODO, muFODO → π / 2} // Simplify]
```

In[116]:=
```
Plot[Evaluate[Simplify[{β_FODO[s]/L_FODO, betaDFODO[s]/L_FODO, D_FODO[s]/(L_FODO φ_FODO)} /. {s → sl L_FODO, μ_FODO → π/2}]],
 {sl, 0, 1}, PlotStyle → {Blue, Red, Darker[Green]},
 FrameLabel → {"s/L_FODO", "β_x/L, β_y/L,D_x/(Lφ) for μ=π/2"}]
```

In[115]:=
```
Manipulate[Plot[
  Evaluate[Simplify[{β_FODO[s]/L_FODO, betaDFODO[s]/L_FODO, D_FODO[s]/(L_FODO φ_FODO)} /. {s → sl L_FODO, μ_FODO → π/2 mudeg/90}]],
  {sl, 0, 1}, PlotStyle → {Blue, Red, Darker[Green]}, PlotRange → {0, 2}, FrameLabel →
  {"s/L_FODO", "β_x/L, β_y/L,D_x/(Lφ)", "μ=" <> ToString[mudeg] <> "°"}], {mudeg, 0, 180}]
```

Out[115]=



■ Dispersion and momentum compaction

Average of dispersion over the FODO cell is

In[117]:=
```
Notation[⟨D_FODO⟩ ⟺ DavFODO];
IntroduceSymbol[DavFODO, "is the average dispersion over the FODO cell.", Meter];
```

In[119]:= $\text{FODODav} = \text{DavFODO} == \frac{1}{1/2}\int_0^{1/2} \text{DFODO}[\text{sl LFODO}]\, d\text{sl}$

Out[119]= $\langle D_{FODO}\rangle == \frac{1}{48}\left(-L_{FODO}\,\varphi_{FODO} + 12\, L_{FODO}\,\varphi_{FODO}\,\text{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2\right)$

Check this agrees with the Handbook formula (7)

In[120]:= $\text{Simplify}\left[\left(\langle D_{FODO}\rangle \text{ /. First@Solve[FODODav, DavFODO]}\right) == \frac{1}{4} L_{FODO}\,\varphi_{FODO}\left(\frac{1}{\text{Sin}\left[\frac{\mu_{FODO}}{2}\right]^2} - \frac{1}{12}\right)\right]$

Out[120]= True

This gives the momentum compaction factor

In[121]:=
```
Notation[α_cFODO ⟺ alphacFODO];
IntroduceSymbol[alphacFODO, "is the momentum compaction factor of a FODO cell.", 1];
```

In[123]:= $\text{FODOalphac} = \text{alphacFODO} == \text{phiFODO}\,\frac{\text{DavFODO}}{\text{LFODO}}$

Out[123]= $\alpha_{cFODO} == \frac{\langle D_{FODO}\rangle\,\varphi_{FODO}}{L_{FODO}}$

In[124]:=
```
FODOOpticalRelations := Join[FODOOpticalFunctions, {FODODav, FODOalphac}]
```

In[125]:=
```
alphacFODO /. First@Solve[FODOOpticalRelations, {alphacFODO, DavFODO}]
```

Out[125]= $\frac{1}{96}\varphi_{\text{FODO}}^2\,(23+\text{Cos}[\mu_{\text{FODO}}])\,\text{Csc}\left[\frac{\mu_{\text{FODO}}}{2}\right]^2$

Check that agrees with Handbook (8):

In[126]:= $\frac{1}{96}\varphi_{\text{FODO}}^2\,(23+\text{Cos}[\mu_{\text{FODO}}])\,\text{Csc}\left[\frac{\mu_{\text{FODO}}}{2}\right]^2 == \frac{L_{\text{FODO}}\,\varphi_{\text{FODO}}\left(\frac{1}{\text{Sin}\left[\frac{\mu_{\text{FODO}}}{2}\right]^2}-\frac{1}{12}\right)\varphi_{\text{FODO}}}{4\,L_{\text{FODO}}}$ // Simplify

Out[126]= True

In[127]:=
```
Plot[Evaluate[alphacFODO/phiFODO^2 /. First@Solve[FODOOpticalRelations, {alphacFODO, DavFODO}] //.
   muFODO → mudeg 2 π/360], {mudeg, 0, 180}, FrameLabel → {"μ/°", "αc/φ^2"}, PlotRange → {0, 5}]
```

Out[127]= 

With the explicit choices of cell length

In[128]:=
```
Plot[Evaluate[alphacFODO /. First@Solve[FODOOpticalRelations, {alphacFODO, DavFODO}] /.
    {fullLHCcell, halfLHCcell} //. muFODO → mudeg 2 π/360],
  {mudeg, 0, 180}, FrameLabel → {"μ/°", "α_c"}, PlotRange → {0, 0.001}]
```

Out[128]=



Chromaticity

In[129]:=
```
Notation[ξ_FODO ⟺ xiFODO];
IntroduceSymbol[xiFODO, "is the natural chromaticity per FODO cell.", 1];
```

In[131]:= FODOchromaticity = $\xi_{FODO} == -\frac{\mathrm{Tan}\left[\frac{\mu_{FODO}}{2}\right]}{\pi}$

Out[131]= $\xi_{FODO} == -\frac{\mathrm{Tan}\left[\frac{\mu_{FODO}}{2}\right]}{\pi}$

In[132]:=
```
Plot[Evaluate[xiFODO /. Solve[FODOchromaticity, xiFODO] /. muFODO → mudeg 2 π/360],
  {mudeg, 0, 180}, FrameLabel → {"μ/°", "ξ_FODO"}]
```

Out[132]=



The specific (per unit tune) or logarithmic chromaticity, neglecting the insertions

```
In[133]:= Plot[Evaluate[xiFODO/muFODO 2 π /. Solve[FODOchromaticity, xiFODO] /. muFODO → mudeg 2 π/360],
  {mudeg, 0, 180}, FrameLabel → {"μ/°", "ξFODO"}]
```

Out[133]=



Plot chromaticity of the whole arc for the two choices of cell length:

```
In[134]:= Plot[Evaluate[
   (Ncell /. {fullLHCcell, halfLHCcell}) xiFODO /. Solve[FODOchromaticity, xiFODO] /.
    muFODO → mudeg 2 π/360], {mudeg, 0, 180}, FrameLabel → {"μ/°", "ξarcs"}]
```

Out[134]=



■ Synchrotron radiation integrals

```
In[135]:= Notation[I2 FODO ⟺ I2FODO];
IntroduceSymbol[I2FODO, "is the synchrotron radiation integral.", Meter^-1];

In[137]:= Notation[I3 FODO ⟺ I3FODO];
IntroduceSymbol[I3FODO, "is the synchrotron radiation integral.", Meter^-2];

In[139]:= FODOI2 = I2FODO == LFODO/rhoFODO^2
```

Out[139]= $I_{2\,\mathrm{FODO}} == \frac{L_{\mathrm{FODO}}}{\rho_{\mathrm{FODO}}^2}$

In[140]:=
```
I2FODO /. Solve[{FODOI2, FODOrho}, {rhoFODO, I2FODO}]
```

Out[140]= $\left\{\frac{L_{FODO}\,\varphi_{FODO}^2}{(L_{FODO} - 5\,\text{Meter} - 2\,L_Q)^2}\right\}$

In[141]:=
```
FODOI3 = I3FODO == LFODO / rhoFODO^3
```

Out[141]= $I_{3\,FODO} == \frac{L_{FODO}}{\rho_{FODO}^3}$

In[142]:=
```
I3FODO /. Solve[{FODOI3, FODOrho}, {rhoFODO, I3FODO}]
```

Out[142]= $\left\{\frac{L_{FODO}\,\varphi_{FODO}^3}{(L_{FODO} - 5\,\text{Meter} - 2\,L_Q)^3}\right\}$

In[143]:=
```
Notation[I5 FODO ⟺ I5FODO];
IntroduceSymbol[I5FODO, "is the synchrotron radiation integral.", Meter^-2];
```

In[145]:= $\text{FODOI5} = I_{5\,FODO} == \frac{\varphi_{FODO}^5\left(1 - \frac{3}{4}\,\text{Sin}\left[\frac{\mu_{FODO}}{2}\right]^2 + \frac{1}{60}\,\text{Sin}\left[\frac{\mu_{FODO}}{2}\right]^4\right)}{4\,L_{FODO}\,\text{Sin}\left[\frac{\mu_{FODO}}{2}\right]^2\,\text{Sin}[\mu_{FODO}]}$

Out[145]= $I_{5\,FODO} == \frac{\varphi_{FODO}^5\,\text{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2\,\text{Csc}[\mu_{FODO}]\left(1 - \frac{3}{4}\,\text{Sin}\left[\frac{\mu_{FODO}}{2}\right]^2 + \frac{1}{60}\,\text{Sin}\left[\frac{\mu_{FODO}}{2}\right]^4\right)}{4\,L_{FODO}}$

In[147]:=
```
Notation[I8 FODO ⟺ I8FODO];
IntroduceSymbol[I8FODO, "is the synchrotron radiation integral.", Meter^-2];
```

$$I_8 = \int D_x{}^2\, K_1(s)^2\, ds$$

In[149]:= $\text{FODOI8} = I_{8\,FODO} == \left(L_Q\,K_{FODO}^2\right)\left((D_x^+)^2 + (D_x^-)^2\right)$ /.
First@Solve[FODOOpticalFunctions, $\{D_x^+, D_x^-, K_{FODO}, f_{FODO}\}$] // Simplify

Out[149]= $I_{8\,FODO} == \frac{\varphi_{FODO}^2\left(1 + 4\,\text{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2\right)}{2\,L_Q}$

In[150]:=
```
FODORadiationIntegrals := {FODOI2, FODOI3, FODOI5, FODOI8}
```

■ Damping times

■ Damping partition variation

In[164]:=
```
FODOrho
```

Out[164]= $\rho_{FODO} == \frac{L_{FODO} - 5\,\text{Meter} - 2\,L_Q}{\varphi_{FODO}}$

In[165]:=
```
Notation[J'ε ⟺ dJedp];
IntroduceSymbol[dJedp,
  "is the derivative of the longitudinal damping partition number with respect to δ.",
  1];
```

In[167]:=
```
FODOdJedp =
  dJedp == (2 I8FODO / I2FODO /. First@Solve[{FODOI2, FODOI8, FODOrho0}, {rhoFODO, I2FODO, I8FODO}])
```

Out[167]= $J'_\epsilon == -\frac{81\, L_{FODO}\, (-9 + \text{Cos}[\mu_{FODO}])\, \text{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2}{200\, L_Q}$

This only depends on the ratio of quad to cell lengths:

In[168]:=
```
FODOdJedp /. Lquad → f_Q LFODO // Simplify
```

Out[168]= $J'_\epsilon == -\frac{81\, (-9 + \text{Cos}[\mu_{FODO}])\, \text{Csc}\left[\frac{\mu_{FODO}}{2}\right]^2}{200\, f_Q}$

In[169]:=
```
Evaluate[
  dJedp /. First@Solve[FODOdJedp, dJedp] /. L_Q → f L_FODO /. muFODO → 2 π mudeg/360 // Simplify]
```

Out[169]= $-\frac{81\left(-9 + \text{Cos}\left[\frac{\text{mudeg}\,\pi}{180}\right]\right)\text{Csc}\left[\frac{\text{mudeg}\,\pi}{360}\right]^2}{200\, f}$

In[170]:=
```
Manipulate[Plot[Evaluate[
   Evaluate[dJedp /. First@Solve[FODOdJedp, dJedp] /. L_Q → f L_FODO /. muFODO → 2 π mudeg/360 //
      Simplify] /. L_Q → f L_FODO /. muFODO → 2 π mudeg/360 // Simplify], {mudeg, 0, 110},
  PlotRange → {0, 1000}, FrameLabel → {"μ/°", "J_ϵ'(δ)", "L_Q/L_FODO=" <> ToString[f]}], {f,
  0.01, .2}]
```

Out[170]=


In[171]:=
```
Table[
  Evaluate[dJedp /. First@Solve[FODOdJedp, dJedp] /. L_Q → f L_FODO /. muFODO → 2 π mudeg/360 // N //
    Simplify], {f, 0.01, .2, 0.04}]
```

```
Plot[Evaluate[Table[
   dJedp /. First@Solve[FODOdJedp, dJedp] /. L_Q → f L_FODO /. muFODO → 2 π mudeg/360 // N // Simplify,
   {f, 0.01, .1, 0.01}]], {mudeg, 0, 110}, PlotRange → {0, 1000},
 FrameLabel → {"μ/°", "J_ϵ'(δ)", "L_Q/L_FODO=" <> ToString[f]}]
```



In[172]:= `{LquadLEP, LFODOLEP, LquadLEP / LFODOLEP}`

Out[172]= `{1.6, 79., 0.0202532}`

J'_ϵ

It might be possible to adjust J'_ϵ with Robinson wigglers - to be studied - if the option of tuning emittance this way is to be retained.

Polarization time

Horizontal emittance

`? EmittanceX`

EmittanceX[Eb,I2,I5,Jx] returns the horizontal emittance of an electron beam of energy Eb in a storage ring with synchrotron integrals I2, I5 and damping partition number Jx; Je has the default value 1.

In[194]:= `EmittanceX[Ee, I2, I5, Jx]`

Out[194]= $$\frac{55\, E_e^2\, I5\, \lambda_e}{64\sqrt{3}\, c^4\, I2\, Jx\, m_e^2\, \pi}$$

In[195]:= `EmittanceX[E_e, I_2 FODO, I_5 FODO, Jx] /. First[Solve[{FODOI5, FODOI2, FODOphi}, {φ_FODO, I_5 FODO, I_2 FODO}]]`

Out[195]= $$\frac{1}{c^4\, Jx\, m_e^2\, \text{Meter}^5} 2.52982\times10^{-51}\, E_e^2\, \lambda_e\, L_{FODO}^3\, \rho_{FODO}^2\, \text{Csc}[\mu_{FODO}]$$
$$\left(-3.89265\times10^{31} + 5.1902\times10^{31}\, \text{Csc}[0.5\, \mu_{FODO}]^2 + 8.65034\times10^{29}\, \text{Sin}[0.5\, \mu_{FODO}]^2\right)$$

Going to a shorter cell length would help.

In[198]:=
```
EmittanceX[Ee, I2FODO, I5FODO, 1] /.
  First@Solve[{FODOI5, FODOI2, FODOrho0}, {I5FODO, I2FODO, phiFODO}]
```

Out[198]=

$$\frac{72171\sqrt{3}\; E_e^2\, \lambda_e\, L_{\text{FODO}}^3\, (303 + 176\, \text{Cos}[\mu_{\text{FODO}}] + \text{Cos}[2\,\mu_{\text{FODO}}])\, \text{Csc}\left[\frac{\mu_{\text{FODO}}}{2}\right]^3 \text{Sec}\left[\frac{\mu_{\text{FODO}}}{2}\right]}{1638400000\, c^4\, m_e^2\, \pi\, \rho_{\text{FODO}}^3}$$

Using the approximation that doesn't include the dependence on quad length:

In[201]:=
```
N[EmittanceX[Ee, I2 FODO, I5 FODO, 1] /.
   First[Solve[{FODOI5, FODOI2, FODOrho0}, {I5 FODO, I2 FODO, ρFODO}]] /. halfLHCcell]
```

Out[201]=

$$\frac{1}{\text{Kilogram}^2\, \text{Meter}^3} 60021.6\, E_e^2\, \text{Second}^4$$
$$(303. + 176.\, \text{Cos}[\mu_{\text{FODO}}] + \text{Cos}[2.\,\mu_{\text{FODO}}])\, \text{Csc}[0.5\,\mu_{\text{FODO}}]^3\, \text{Sec}[0.5\,\mu_{\text{FODO}}]$$

In[208]:=
```
Plot[Evaluate[N[EmittanceX[Ee, I2 FODO, I5 FODO, 1] / Nano Meter /.
      First[Solve[{FODOI5, FODOI2, FODOrho0}, {I5 FODO, I2 FODO, ρFODO}]] /. halfLHCcell /.
    muFODO → 2 π/360 mudeg /. Ee → 70 GeV] // ToFundamentalSI],
 {mudeg, 40, 120}, FrameLabel → {"μ/°", "εx/nm", "Emittance at 70 GeV"}]
```

Out[208]=



Including the quad length dependence:

In[219]:=
```
N[EmittanceX[Ee, I2 FODO, I5 FODO, 1] /.
   First[Solve[{FODOI5, FODOI2, FODOrho, FODOphi}, {I5 FODO, I2 FODO, ρFODO}]] /. halfLHCcell]
```

Out[219]=

$$\frac{1}{\text{Kilogram}^2\, \text{Meter}^5} 1.28196\times 10^{-13}\, E_e^2\, \text{Second}^4\, \text{Csc}[\mu_{\text{FODO}}]$$
$$\left(-45. + 60.\, \text{Csc}[0.5\,\mu_{\text{FODO}}]^2 + \text{Sin}[0.5\,\mu_{\text{FODO}}]^2\right) \left(2.75665\times 10^9\, \text{Meter} - 1.1379\times 10^8\, L_Q\right)^2$$

In[220]:= `% /. L_Q → 0`

Out[220]= $$\frac{1}{\text{Kilogram}^2\,\text{Meter}^5}1.28196\times10^{-13}\,E_e^2\,\text{Second}^4\,\text{Csc}[\mu_{\text{FODO}}]\left(-45.+60.\,\text{Csc}[0.5\,\mu_{\text{FODO}}]^2+\text{Sin}[0.5\,\mu_{\text{FODO}}]^2\right)\left(2.75665\times10^9\,\text{Meter}-1.1379\times10^8\,L_Q\right)^2$$

In[221]:= `Simplify[ToFundamentalSI[% /. E_e → 70 GeV]]`

Out[221]= $$\frac{1}{\text{Meter}}1.61247\times10^{-29}\,\text{Csc}[\mu_{\text{FODO}}]\left(-45.+60.\,\text{Csc}[0.5\,\mu_{\text{FODO}}]^2+\text{Sin}[0.5\,\mu_{\text{FODO}}]^2\right)\left(2.75665\times10^9\,\text{Meter}-1.1379\times10^8\,L_Q\right)^2$$

In[222]:= `% /. muFODO → 2 π 90 / 360`

Out[222]= $$\frac{1.21741\times10^{-27}\left(2.75665\times10^9\,\text{Meter}-1.1379\times10^8\,L_Q\right)^2}{\text{Meter}}$$

By either calculation, this is close to the required emittance (7-8 nm) to match the LHC protons in the EPAC 2008 paper. A final adjustment could be made by changing the β functions (rematch of the IR) or perhaps by damping partition variation. We assume we don't want to use wigglers at high energy.

We could use a (90°,60°) optics with this cell length. It should have good dynamic aperture and be insensitive to imperfections (LEP experience with this choice).

Power and luminosity

Wigglers

LEP had 20 wigglers in 3 different classes, used operationally as transparent knobs (in combination with adjustments to nearby quadrupoles to match edge-focusing.

It is probably desirable to have damping wigglers in the LHeCe ring at injection.

Aperture of the arcs

Energy sawtooth estimate. Fractional momentum deviation can reach 1% since there is just one RF station

```
dpmax = 0.01
```

0.01

```
FODODxp
```

$$D_x^{\prime} == \frac{1}{4}\,L_{\text{FODO}}\,\varphi_{\text{FODO}}\,\text{Csc}\left[\frac{\mu_{\text{FODO}}}{2}\right]^2\left(1+\frac{1}{2}\,\text{Sin}\left[\frac{\mu_{\text{FODO}}}{2}\right]\right)$$

```
Solve[FODODxp]
```

$$\left\{\left\{D_x^{\prime}\to\frac{1}{8}\,L_{\text{FODO}}\,\varphi_{\text{FODO}}\,\text{Csc}\left[\frac{\mu_{\text{FODO}}}{2}\right]^2\left(2+\text{Sin}\left[\frac{\mu_{\text{FODO}}}{2}\right]\right)\right\}\right\}$$

```
Dxp dpmax /. First@Solve[FODODxp] /. halfLHCcell /. muFODO → π / 2
```

0.00617643 Meter

This must be included in the aperture budget for the arcs

```
Dxp dpmax + 15 √(betap Ex + Dxp² sige²) /. First@Solve[{FODODxp, FODObetap}, {Dxp, betap}] /.
   halfLHCcell /. {muFODO → π / 2, Ex → 10 Nano Meter, sige → 10^-3} // ToFundamentalSI
```

0.0232393 Meter

```
LEPvacuumChamber
```

{65.5 Meter Milli, 35. Meter Milli}

We can probably make do with a smaller half-aperture than LEP, perhaps ±40 mm in the horizontal plane, a reduction of 30%.

This will reduce the size of magnets, etc. to be evaluated in more detail.

With just one beam in the ring, it might be possible to reduce the aperture further with a scheme of orbit corrector magnets to cancel the orbit sawtooth.

Revolution-frequency matching and energy range

See other notebook!

In particular for Pb

RF voltage

Possible LEP-III option ?

Luminosity estimate for e-Pb collisions

This went into EPAC2008 paper.

```
LHeCPbBeam = {Z → 82, A → 208, m_ion → 1.0072756064562605` AMU, Nbh → 7. 10^7, Ebh → Z 7. TeV,
   kb -> f0 → c / (26 658.883 Meter), Exh → 0.501 Nano Meter,  Eyh → 0.501 Nano Meter,
   bxstarh → 1.8 Meter, bystarh → 0.5 Meter, fb → 40. MHz, gammah -> Ebh / (mion c²)}
```

$\{Z \to 82,\ A \to 208,\ m_{ion} \to 1.00728\ \text{AMU},\ Nbh \to 7. \times 10^{7},\ Ebh \to 7.\ \text{ElectronVolt Tera}\ Z,$
$k_b \to f_0 \to \frac{0.0000375109\ c}{\text{Meter}},\ Exh \to 0.501\ \text{Meter Nano},\ Eyh \to 0.501\ \text{Meter Nano},$
$bxstarh \to 1.8\ \text{Meter},\ bystarh \to 0.5\ \text{Meter},\ fb \to 40.\ \text{MHz},\ gammah \to \frac{Ebh}{c^2\ m_{ion}}\}$

```
LHeCeBeam = {Ee → 70. GeV, f0 → c / (26 658.883 Meter),
   kbe → 592, Nbe → 1.4 10^10, Exe → 7.6 Nano Meter,  Eye → 3.8 Nano Meter,
   bxstare → 12.7 Centi Meter,  bystare → 7.1 Centi Meter, fb → 40. Mega Hertz}
```

$\{Ee \to 70.\ \text{ElectronVolt Giga},\ f_0 \to \frac{0.0000375109\ c}{\text{Meter}},$
$kbe \to 592,\ Nbe \to 1.4 \times 10^{10},\ Exe \to 7.6\ \text{Meter Nano},\ Eye \to 3.8\ \text{Meter Nano},$
$bxstare \to 12.7\ \text{Centi Meter},\ bystare \to 7.1\ \text{Centi Meter},\ fb \to 40.\ \text{Hertz Mega}\}$

```
lumiRule = lumihe → kbe Nbe f0 Nbh / (4 π Exh Sqrt[bxstarh bystarh])
```

$$\text{lumihe} \to \frac{f_0 \text{ kbe Nbe Nbh}}{4 \sqrt{\text{bxstarh bystarh}} \text{ Exh } \pi}$$

The electron-nucleus luminosity would b e

```
Convert[lumihe /. lumiRule //. LHeCPbBeam //. LHeCeBeam // N, lumiUnits]
```

$$\frac{1.09234 \times 10^{29}}{\text{Centi}^2 \text{ Meter}^2 \text{ Second}}$$

The corresponding electron-nucleon luminosity

```
208 %
```

$$\frac{2.27207 \times 10^{31}}{\text{Centi}^2 \text{ Meter}^2 \text{ Second}}$$

This is with the same electron current/bunch as ep. For the same RF power, one can increase this by

```
592 / 2800 //
```

4.72973

```
(10 / 50.)^(1/4)
```

0.66874

```
% 70
```

46.8118